MLC Lab Visit - Lab 07 - Maple

Mth 355 (a.k.a. Mth 399) Feb 19, 2003 Maple 7
Bent E. Petersen

petersen@math.orst.edu

There are 5 problems below. Problem solutions are due Feb 26, 2003. Email your solutions to me as Maple worksheet attachments. Your worksheet must execute correctly for full credit.

In this week's lab we investigate a few randomly chosen features of Maple.

Interpolation Polynomials

```
> restart;
> with(plots):
Warning, the name changecoords has been redefined
```

Example 1

Maple provides a builtin command interp() for computing interpolation polynomials.

```
> poly1:=interp([1,3,4,2],[2,1,3,1],z);
```

$$poly1 := \frac{1}{6} z^3 - \frac{1}{2} z^2 - \frac{2}{3} z + 3$$

The first parameter we pass to interp() is the list of (distinct) abscissas, the second is the list of ordinates and the third is a name, the name for the variable to be used in the polynomial.

If you want a polynomial function rather than a polynomial expression in some variable, you can use unapply():

```
> poly1fun:=unapply(interp([1,3,4,2],[2,1,3,1],z),z);
```

$$poly1fun := z \rightarrow \frac{1}{6} z^3 - \frac{1}{2} z^2 - \frac{2}{3} z + 3$$

Let's check that it worked:

```
> poly1fun(1); poly1fun(3); poly1fun(4); poly1fun(2);
```

$$2$$

$$1$$
$$3$$
$$1$$

Example 2

If you have a list of points you want to interpolate you can extract the abscissas and ordinates by using the sequence comamnd seq().

```
> data2:=[ [1,2], [2,-1], [3,-2], [-1,1], [-2,7], [8,6], [7,5] ];
```

$$data2 := [[1, 2], [2, -1], [3, -2], [-1, 1], [-2, 7], [8, 6], [7, 5]]$$

```
> XX2:=[seq(data2[k][1],k=1..nops(data2))];
```

$$XX2 := [1, 2, 3, -1, -2, 8, 7]$$

```
> YY2:=[seq(data2[k][2],k=1..nops(data2))];
```

$$YY2 := [2, -1, -2, 1, 7, 6, 5]$$

Let's construct the interpolation polynomial and plot it together with the data:

```
> poly2:=interp(XX2,YY2,z);
```

$$poly2 := \frac{79}{16800} z^6 - \frac{521}{6048} z^5 + \frac{23567}{50400} z^4 - \frac{2435}{6048} z^3 - \frac{24403}{12600} z^2 + \frac{1495}{1512} z + \frac{667}{225}$$

```
> dataplot2:=plot(data2,style=point,symbol=circle,symbolsize=16,c
olor=blue):
> polyplot2:=plot(poly2,z=1..7,color=red):
```

Here's the plot:

```
> display([dataplot2,polyplot2]);
```



Example 3

A convenient way to construct an interpolation polynomial for a function is to use the map() command to evaluate the function at each abscissa. Let's consider the function 1/(1+16x^2) on [-1,1].

```
> XX3:=[seq(-1+k/6,k=0..12)];
```

$$XX3 := \left[-1, \frac{-5}{6}, \frac{-2}{3}, \frac{-1}{2}, \frac{-1}{3}, \frac{-1}{6}, 0, \frac{1}{6}, \frac{1}{3}, \frac{1}{2}, \frac{2}{3}, \frac{5}{6}, 1\right]$$

```
> YY3:=map(x->(1/(1+16*x^2)),XX3);
```

$$YY3 := \left[\frac{1}{17}, \frac{9}{109}, \frac{9}{73}, \frac{1}{5}, \frac{9}{25}, \frac{9}{13}, 1, \frac{9}{13}, \frac{9}{25}, \frac{1}{5}, \frac{9}{73}, \frac{9}{109}, \frac{1}{17}\right]$$

Note the use of an anonymous function in the argument of map().

We will need a list of points to plot the original data:

```
> data3:=[seq([XX3[k],YY3[k]],k=1..nops(XX3))];
```

$$data3 := \left[\left[-1, \frac{1}{17}\right], \left[\frac{-5}{6}, \frac{9}{109}\right], \left[\frac{-2}{3}, \frac{9}{73}\right], \left[\frac{-1}{2}, \frac{1}{5}\right], \left[\frac{-1}{3}, \frac{9}{25}\right], \left[\frac{-1}{6}, \frac{9}{13}\right], [0, 1], \left[\frac{1}{6}, \frac{9}{13}\right], \left[\frac{1}{3}, \frac{9}{25}\right],\right.$$

$$\left[\frac{1}{2}, \frac{1}{5}\right], \left[\frac{2}{3}, \frac{9}{73}\right], \left[\frac{5}{6}, \frac{9}{109}\right], \left[1, \frac{1}{17}\right]\right]$$

```
> poly3:=evalf(interp(XX3,YY3,t),6);
```

$$poly3 := 500.770\, t^{12} - 1297.13\, t^{10} + 1241.42\, t^{8} - 554.928\, t^{6} + 123.022\, t^{4} - 14.0919\, t^{2} + 1.$$

```
> dataplot3:=plot(data3,style=point,symbol=circle,symbolsize=16,color=blue):
> polyplot3:=plot(poly3,t=-1..1,color=red,thickness=2):
```

Note the previous example shows one way of plotting two functions on one graph.

```
> display({polyplot3,dataplot3});
```



Notice how the polynomial oscillates too much in some sense. For example, the original function is positive on [-1,1] but the interpolation polynomial is far from positive.

```
>
```

Interpolation Splines

```
> restart;
> with(plots):
```

Warning, the name changecoords has been redefined

Maple computes splines of all degrees. Here we will look only at (natural) cubic splines. The parameters are much the same as for interp(), but the abscissas must be in increasing order.

```
> XX:=[seq(-1+k/6,k=0..12)];
```

$$XX := \left[-1, \frac{-5}{6}, \frac{-2}{3}, \frac{-1}{2}, \frac{-1}{3}, \frac{-1}{6}, 0, \frac{1}{6}, \frac{1}{3}, \frac{1}{2}, \frac{2}{3}, \frac{5}{6}, 1\right]$$

```
> YY:=map(x->(1/(1+16*x^2)),XX);
```

$$YY := \left[\frac{1}{17}, \frac{9}{109}, \frac{9}{73}, \frac{1}{5}, \frac{9}{25}, \frac{9}{13}, 1, \frac{9}{13}, \frac{9}{25}, \frac{1}{5}, \frac{9}{73}, \frac{9}{109}, \frac{1}{17}\right]$$

```
> sp:=evalf(spline(XX,YY,x,cubic),6);
```

$$sp := \begin{cases} .802983 + 1.98192\,x + 1.85664\,x^2 + .618879\,x^3 & x < -.833333 \\ .775796 + 1.88404\,x + 1.73919\,x^2 + .571899\,x^3 & x < -.666667 \\ .962438 + 2.72393\,x + 2.99902\,x^2 + 1.20182\,x^3 & x < -.500000 \\ 1.41676 + 5.44986\,x + 8.45087\,x^2 + 4.83638\,x^3 & x < -.333333 \\ 1.18878 + 3.39806\,x + 2.29548\,x^2 - 1.31901\,x^3 & x < -.166667 \\ 1. - 18.0929\,x^2 - 42.0957\,x^3 & x < 0. \\ 1. - 18.0929\,x^2 + 42.0957\,x^3 & x < .166667 \\ 1.18878 - 3.39806\,x + 2.29548\,x^2 + 1.31901\,x^3 & x < .333333 \\ 1.41676 - 5.44986\,x + 8.45087\,x^2 - 4.83638\,x^3 & x < .500000 \\ .962438 - 2.72393\,x + 2.99902\,x^2 - 1.20182\,x^3 & x < .666667 \\ .775796 - 1.88404\,x + 1.73919\,x^2 - .571899\,x^3 & x < .833333 \\ .802983 - 1.98192\,x + 1.85664\,x^2 - .618879\,x^3 & otherwise \end{cases}$$

```
> pts:=[seq([XX[k],YY[k]],k=1..nops(XX))];
```

$$pts := \left[\left[-1, \frac{1}{17}\right], \left[\frac{-5}{6}, \frac{9}{109}\right], \left[\frac{-2}{3}, \frac{9}{73}\right], \left[\frac{-1}{2}, \frac{1}{5}\right], \left[\frac{-1}{3}, \frac{9}{25}\right], \left[\frac{-1}{6}, \frac{9}{13}\right], [0, 1], \left[\frac{1}{6}, \frac{9}{13}\right], \left[\frac{1}{3}, \frac{9}{25}\right], \left[\frac{1}{2}, \frac{1}{5}\right], \left[\frac{2}{3}, \frac{9}{73}\right], \left[\frac{5}{6}, \frac{9}{109}\right], \left[1, \frac{1}{17}\right]\right]$$

```
> dataplot:=plot(pts,style=point,symbol=circle,symbolsize=16,color=blue):
> splnplot:=plot(sp,x=-1..1,color=red):
> display({dataplot,splnplot});
```



Notice how much better the interpolation spline tracks the data than the interpolation polynomial does.

```
>
```

Planar Graphs

```
> restart;
> with(networks):
```

We discussed planar graphs in class. According to Kuratowski (circa 1930) a graph is planar if and only if it contains no subgraph that is homeomorphic to K5 or K3_3. Note the term "homeomorphic" is used here in a sense peculiar to graph theory: two graphs are homeomorphic if they can be obtained from the same graph by "bisecting" some of its edges. This is a beautiful result though I have no idea how useful it is.

Maple does have a builtin planarity test, isplane().

```
> G1:=icosahedron():
> isplanar(G1);
```

true

```
> draw(G1);
```



It is hard to believe that G1 is planar!

```
> G2:=petersen():
> isplanar(G2);
```

false

```
> draw(G2);
```




```
> K3_3:=complete(3,3):
> isplanar(K3_3);
```

false

```
> draw(K3_3);
```



```
> K5:=complete(5):
> isplanar(K5);
```

false

```
> draw(K5);
```



The Maple default labeling of the edges of K5 is

```
> edges(K5);
```

$\{e2, e9, e3, e4, e5, e1, e10, e7, e6, e8\}$

Once we know the labels we can remove an edge. Note the edge is removed from K5 directly - the new graph replaces the old, so there is no point in assigning the result.

```
> delete(e1,K5):
> draw(K5);
```




```
> G3:=cube(3):
> isplanar(G3);
```

true

```
> draw(G3);
```



```
> G4:=cube(4):
> isplanar(G4);
```

false

```
> draw(G4);
```




```
> G5:=complete(2,3):
> isplanar(G5);
```

true

```
> draw(G5);
```




```
>
```

Numeric Derivative Estimates

```
> restart;
```

Here is a silly example, but it illustrates a general procedure (method of undetermined coefficients).

Suppose we wish to have an estimate of the second derivative of f at a in terms of f(a), f(a+h) and f(a+3h):

```
> g:=h->A1*f(a)+A2*f(a+h)+A3*f(a+3*h);
```

$$g := h \rightarrow A1\ \mathrm{f}(a) + A2\ \mathrm{f}(a+h) + A3\ \mathrm{f}(a+3\,h)$$

We expand g in a Taylor polynomial with center at h=0

```
> expr:=taylor(g(h),h=0,4);
```

$$expr := (A1\ \mathrm{f}(a) + A2\ \mathrm{f}(a) + A3\ \mathrm{f}(a)) + (3\,A3\ \mathrm{D}(f)(a) + A2\ \mathrm{D}(f)(a))\,h + \left(\frac{9}{2}A3\,(\mathrm{D}^{(2)})(f)(a) + \frac{1}{2}A2\,(\mathrm{D}^{(2)})(f)(a)\right)h^2 + \left(\frac{9}{2}A3\,(\mathrm{D}^{(3)})(f)(a) + \frac{1}{6}A2\,(\mathrm{D}^{(3)})(f)(a)\right)h^3 + \mathrm{O}(h^4)$$

We want the first and second coefficients here to be 0 and the third to be 1.

```
> for k from 0 to 3 do coef[k]:=coeff(expr,h,k); od;
```

$$coef_0 := A1\ f(a) + A2\ f(a) + A3\ f(a)$$

$$coef_1 := 3\ A3\ D(f)(a) + A2\ D(f)(a)$$

$$coef_2 := \frac{9}{2} A3\ (D^{(2)})(f)(a) + \frac{1}{2} A2\ (D^{(2)})(f)(a)$$

$$coef_3 := \frac{9}{2} A3\ (D^{(3)})(f)(a) + \frac{1}{6} A2\ (D^{(3)})(f)(a)$$

```
> soln:=solve({coef[0]=0,coef[1]=0,coef[2]=(D@@2)(f)(a)},{A1,A2,A
  3});
```

$$soln := \{A3 = \frac{1}{3}, A2 = -1, A1 = \frac{2}{3}\}$$

```
> expr2:=subs(soln,expr);
```

$$expr2 := (D^{(2)})(f)(a)\ h^2 + \frac{4}{3} (D^{(3)})(f)(a)\ h^3 + O(h^4)$$

To seperate out the second derivative we need to divide by h^2. So here’s our expression

```
> est:=subs(soln,g(h)/h^2);
```

$$est := \frac{\frac{2}{3} f(a) - f(a+h) + \frac{1}{3} f(a+3h)}{h^2}$$

We can estimate the error

```
> taylor((D@@2)(f)(a)-est,h=0,4);
```

$$-\frac{4}{3} (D^{(3)})(f)(a)\ h + O(h^2)$$

So, only first order (unless the third derivative at a is zero).

```
>
```

Problems

Problem 1

The doudecahedron graph was discussed in clas. It has 30 edges and 20 vertices each of degree 3. Enter the definition of the doudecahedron into Maple and verify its (obvious of course) planarity.

Problem 2

If we remove an edge from the complete graph K5 is the resulting graph planar?

Problem 3

If we remove an edge from the bipartite graph K3_3 = complete(3,3) is the resulting graph planar?

Problem 4

Find a numeric estimate for the first derivative of f at a in terms of f(a-h), f(a) and f(a+h) of order 2. What particularly nice property do you observe?

Problem 5

Find a numeric estimate for the third derivative of f at a, of as high order as possible, in terms of f(a-h), f(a), f(a+h), f(a+2h) and f(a+3h). What is the order?

>

>

>